
Defining Multicollinearity
What is Multicollinearity?

• Definition
• A statistical phenomenon wherein there exists a perfect or exact 

relationship between predictor variables
• From a conventional standpoint:

• Predictors are highly correlated
• Predictors are co-dependent

• Notes
• When things are related, we say they are linearly dependent

• Fit well into a straight regression line that passes through many data points
• Multicollinearity makes it difficult to come up with reliable estimates of 

individual coefficients for the predictor variables
• Results in incorrect conclusions about the relationship between outcome and predictor 

variables

Presenter
Presentation Notes
For multiple regression model, absence of multicollinearity is essential!

Defining Multicollinearity
What is Multicollinearity?

• Consider multiple linear regression equation:
𝑌𝑌 = 𝑋𝑋β + ε

• Considering Equation:
• Multicollinearity inflates the variances of the parameter 

estimates
• (1) Lack of statistical significance of individual predictor variables, though overall 

model is still significant
• (2) Biased outcome

• The presence of multicollinearity can cause serious problems 
with the estimation of β and its interpretation



SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

#SASGF

Defining Multicollinearity
Why Should We Care About Multicollinearity?

• Problems in Explanation vs Prediction Models
• Explanation:

• More difficult to achieve significance of collinear parameters

• Prediction:
• if estimates are statistically significant, they are only as reliable as any other variable in the model
• If they are not significant, the sum of the coefficient is likely to be reliable

• Corrections:
• In the case of a predictive model: just need to increase sample size
• In the case of an explanatory model: further measures are needed

• Primary concern: as the degree of multicollinearity increases…
• Regression model estimates of the coefficients become unstable
• Standard errors for the coefficients become wildly inflated

Detecting Multicollinearity
Ways to Detect Multicollinearity

• There are three ways to detect multicollinearity
• Examination of the correlation matrix
• Variance Inflation Factor (VIF)
• Eigensystem Analysis of Correlation Matrix

Detecting Multicollinearity
Examination of the Correlation Matrix

• Examination of the Correlation Matrix
• Large correlation coefficients in the  correlation matrix of 

predictor variables indicate multicollinearity
• If there is multicollinearity between any two predictor 

variables, then the correlation coefficient between those 
two  variables will be near to unity

• Proc Corr

Detecting Multicollinearity
Variance Inflation Factor / Tolerance

• Variance Inflation Factor
• The Variance Inflation Factor (VIF) quantifies the severity of 

multicollinearity in an ordinary least-squares regression analysis
• The VIF is an index which measures how much variance of an 

estimated regression coefficient is increased because of 
multicollinearity

• Note: If any of the VIF values exceeds 5 or 10 it implies that the 
associated regression coefficients are poorly estimated because 
of multicollinearity

• Tolerance
• Represented by 1/VIF

Detecting Multicollinearity
Eigensystem Analysis of Correlation Matrix

• Eigensystem Analysis of Correlation Matrix
• The eigenvalues can also be used to measure the presence 

of multicollinearity
• If multicollinearity is present in the predictor variables one 

or more of the eigenvalues will be small (near to zero)
• Note: if one or more of the eigenvalues are small (close to 

zero) and a corresponding condition number is large, then it 
indicates multicollinearity

Detecting Multicollinearity
Example

• Model:
• Suicidal Ideation = Lifetime Substance Use + Age + Gender + 

Racial Identification + Depression + Recent Substance Use + 
Victim of Violence + Participant in Violence

• Suicidal Ideation = β 0 + β 1(Lifetime Substance Use) + 
β 2(Age) + β 3(Gender) + β 4(Racial Identification) + 
β 5(Depression) + β 6(Recent Substance Use) + β 7(Victim of 
Violence) + β 8(Participant in Violence)

Detecting Multicollinearity
Example

• Descriptive Statistics and Initial Examination

/* Building of Table 1: Descriptive and Univariate Statistics */

proc freq data=YRBS_Total;   tables SubAbuseBin_Cat * SI_Cat;  run;

proc freq data=YRBS_Total;   tables (SubAbuse_Cat Age_Cat Sex_Cat Race_Cat Depression_Cat RecSubAbuse_Cat VictimViol_Cat ActiveViol_Cat) * SI_Cat / 
chisq;   run;

data newYRBS_Total (keep =  SubAbuse SubAbuse_Cat Age Age_Cat Sex Sex_Cat Race Race_Cat Depression Depression_Cat RecSubAbuse RecSubAbuse_Cat
VictimViol VictimViol_Cat ActiveViol ActiveViol_Cat SI SI_Cat SubAbuseBin_Cat);  set YRBS_Total (where=  (  (SubAbuse in (0,1,2,3)) and (Age 
in(12,13,14,15,16,17,18)) and (Sex in (1,2)) and (Race in (1,2,3,4,5,6)) and (Depression in (0,1)) and (RecSubAbuse in (0,1)) and (VictimViol in 
(0,1,2)) and (ActiveViol in (0,1,2)) and (SI in (0,1)) and (SubAbuseBin in (0,1)) ));  run;

proc freq data=newYRBS_Total; tables ( Age_Cat Sex_Cat Race_Cat Depression_Cat RecSubAbuse_Cat VictimViol_Cat ActiveViol_Cat ) * SubAbuse_Cat / 
chisq;   run;

/* Building of Table 2: Descriptive and Univariate Statistics */

proc freq data=newYRBS_Total;  tables (SubAbuse_Cat Age_Cat Sex_Cat Race_Cat Depression_Cat RecSubAbuse_Cat VictimViol_Cat ActiveViol_Cat) * SI_Cat / 
chisq;    run;

/* Building of Table 3: Multivariable Logistic Regression w/ Multiplicative Interaction */

proc logistic data = newYRBS_Total;   class SI_Cat (ref='No')  SubAbuse_Cat (ref='1 None')    / param=ref;   model SI_Cat = SubAbuse_Cat / lackfit
rsq;    title 'Suicidal Ideation by Lifetime Substance Abuse Severity, Unadjusted‘;   run;

proc logistic data = newYRBS_Total;    class SI_Cat(ref='No')  SubAbuse_Cat (ref='1 None') Age_Cat (ref='12 or younger') Sex_Cat (ref='Female') 
Race_Cat (ref='White') Depression_Cat (ref='No') RecSubAbuse_Cat (ref='No') VictimViol_Cat (ref='None') ActiveViol_Cat (ref='None') / param=ref;  
model SI_Cat = SubAbuse_Cat Age_Cat Sex_Cat Race_Cat Depression_Cat RecSubAbuse_Cat VictimViol_Cat ActiveViol_Cat / lackfit rsq; title 'Suicidal 
Ideation by Lifetime Substance Abuse Severity, Adjusted - Multivariable Logistic Regression'; run;

Detecting Multicollinearity
Example

Detecting Multicollinearity
Example

• Test: Examination of the Correlation Matrix

/* Examination of the Correlation Matrix */

proc corr data=newYRBS_Total;

var SI SubAbuse Age Sex Race Depression RecSubAbuse VictimViol
ActiveViol;

title 'Suicidal Ideation Predictors - Examination of Correlation 
Matrix';

run;

Detecting Multicollinearity
Example

• Note: No highly correlated predictor variables

Detecting Multicollinearity
Example

• Tests:
• Variance Inflation Factor
• Eigensystem Analysis of Correlation Matrix

/* Multicollinearity Investigation of VIF and Tolerance */

proc reg data=newYRBS_Total;

model SI = SubAbuse Age Sex Race Depression RecSubAbuse VictimViol ActiveViol / vif
tol collin;

title 'Suicidal Ideation Predictors - Multicollinearity Investigation of VIF and 
Tol';

run;
quit;

• Note:
• Common cut point for VIF = 10 (higher indicates multicollinearity)
• Common cut point for Tol = .1 (lower indicates multicollinearity)

Detecting Multicollinearity
Example

• Note: VIF cut point = 10, Tolerance cut point = .1

Detecting Multicollinearity
Example

• Note:
• Eigensystem Analysis of Covariance: If one or more of the 

eigenvalues are small (close to zero) and the corresponding 
condition number is large, then it indicates multicollinearity

Detecting Multicollinearity
Example

• The dataset: SAS Sample Data
libname health "C:\Program Files\SASHome\SASEnterpriseGuide\7.1\Sample\Data";

data health;      set health.lipid;       run;

proc contents data=health; 

title 'Health Dataset with High Multicollinearity'; run; 

• The example: 
• Outcome: Cholesterol loss between baseline and check-up
• Predictors (Baseline): Age, Weight, Cholesterol, Triglycerides, HDL, LDL, Height

Detecting Multicollinearity
Example

• Test: Examination of the Correlation Matrix

/* Assess Pairwise Correlations of Continuous Variables */

proc corr data=health;

var age weight cholesterol triglycerides hdl ldl height;

title 'Health Predictors - Examination of Correlation Matrix';

run;

Detecting Multicollinearity
Example

• Tests:
• Variance Inflation Factor
• Eigensystem Analysis of Correlation Matrix

/* Multicollinearity Investigation of VIF and Tolerance */
proc reg data=health;

model cholesterolloss = age weight cholesterol triglycerides hdl ldl height / 
vif tol collin;
title 'Health Predictors - Multicollinearity Investigation of VIF and Tol'; 

run;

• Note:
• Common cut point for VIF = 10 (higher indicates multicollinearity)
• Common cut point for Tol = .1 (lower indicates multicollinearity)

Detecting Multicollinearity
Example

• Note: VIF cut point = 10, Tolerance cut point = .1

Detecting Multicollinearity
Example

• Eigensystem Analysis of Covariance: If one or more of the eigenvalues are small 
(close to zero) and the corresponding condition number is large, then it indicates 
multicollinearity

Combating Multicollinearity
What Can We Do?

• Easiest
• Drop one or several predictor variables in order to lessen 

the multicollinearity
• If none of the predictor variables can be dropped, 

alternative methods of estimation need to be 
employed:
• Principal Component Regression
• Regularization Techniques

• L1: Lasso Regression
• L2: Ridge Regression

Combating Multicollinearity
Principal Component Regression

• Logic:
• Every linear regression model can be restated in terms of a set of 

orthogonal explanatory variables
• These new variables are obtained as linear combinations of the original 

explanatory variables 
• Often referred to as: Principal Components

• The principal component regression approach combats multicollinearity by 
using less than the full set of principal components in the model

• Calculation:
• To obtain the principal components estimators

• Assume the regressors are arranged in order of decreasing eigenvalues, ʎ1 ≥ ʎ2 ………… ≥ ʎp > 0

• In principal components regression, the principal components 
corresponding to near zero eigenvalues are removed from the analysis
• Least squares is then applied to the remaining components

Combating Multicollinearity
Principal Component Regression Example

/* Principal Component Regression Example */

proc princomp data=health

out=pchealth prefix=z outstat=PCRhealth;

var age weight cholesterol triglycerides hdl ldl height skinfold 
systolicbp diastolicbp exercise coffee;

title 'Health - Principal Component Regression Calculation';

run;

Combating Multicollinearity
Principal Component Regression Example

Principal Component Regression Example

Combating Multicollinearity
Principal Components Regression Example

• Two ways to estimate the appropriate eigenvalue 
cut-off
• Common: cut-off of 1

• Explains at least 1 variable’s worth of information

• Parallel Analysis Criterion
• Eigenvalue obtained for the Nth factor should be larger than the associated 

eigenvalue computed analyzing a set of random data

Combating Multicollinearity
Principal Component Regression Example

• First Example: Common method using eigenvalue of 
at least 1.0000

Combating Multicollinearity
Principal Component Regression Example

• Model is then rewritten in the form of principal 
components:
• Cholesterol Loss = α1z1 + α2z2 + α3z3 + α4z4 + α5z5 + ε

• Zn = Eigenvector(age) + Eigenvector(weight) + …….. + Eigenvector(coffee)
• Estimated values of alphas can be obtained by regression cholesterol loss 

against z1, z2, z3, z4, & z5

/* With Eigenvalue Cutoff of 1.0000 */

proc reg data=pchealth;

model cholesterolloss = z1 z2 z3 z4 z5 / VIF;

title 'Health - Principal Component Regression Adjustment';

run;

Combating Multicollinearity
Principal Components Regression Example

• Second Example: Parallel Analysis Criterion
/****************** Parallel Analysis Program ************************/

/* Location: https://people.ok.ubc.ca/brioconn/nfactors/parallel.sas */

Combating Multicollinearity
Principal Component Regression Example

• Model is then rewritten in the form of principal 
components:
• Cholesterol Loss = α1z1 + α2z2 + α3z3 + ε

• Zn = Eigenvector(age) + Eigenvector(weight) + …….. + Eigenvector(coffee)
• Estimated values of alphas can be obtained by regression cholesterol loss 

against z1, z2, & z3

/* After Parallel Analysis */

proc reg data=pchealth;

model cholesterolloss = z1 z2 z3 / VIF;

title 'Health - Principal Component Regression Adjustment';

run;

Combating Multicollinearity
Principal Component Regression Example

Combating Multicollinearity
Regularization Methods

• Logic:
• Regularization adds a penalty to model parameters (all 

except intercepts) so the model generalizes the data instead 
of overfitting (a side effect of multicollinearity)

• Two main types:
• L1 – Lasso Regression
• L2 – Ridge Regression

Presenter
Presentation Notes
The key difference between these two types of regularization can be found in how they handle the penalty

Combating Multicollinearity
Regularization Methods

• Ridge Regression
• Squared magnitude of the coefficient is added as penalty to loss function

• ∑i=1n (Yi − ∑j=1
p Xijβj)

2
+ ʎ∑j=1

p βj2

• Lasso Regression
• Absolute value of magnitude of the coefficient is added as penalty to loss function

• ∑i=1n (Yi − ∑j=1
p Xijβj)

2
+ ʎ∑j=1

p |βj|

• Result:
• if ʎ =0 then the equation will go back to OLS estimations
• If ʎ is very large, too much weight would be added = under-fitting
• NOTE: need to be careful with choice of ʎ

Presenter
Presentation Notes
Ridge Regression
if lambda (ʎ - the penalty) is zero then the equation will go back to ordinary least squares estimations, whereas a very large lambda would add too much weight to the model which will lead to under-fitting

Lasso Regression
Least Absolute Shrinkage and Selection Operator

Combating Multicollinearity
Regularization Methods

• Key difference:
• Lasso Regression is meant to shrink the coefficient of the less 

important variables to zero
• This works well if feature selection is the goal
• Not necessarily good for multicollinearity

• Ridge Regression adjust weights of the variables
• Goal is not to shrink the coefficients to zero, but to adjust for representation of all 

relevant variables

• Ridge Regression Trade-Off
• We are still dealing with an adjustment
• Naturally results in biased outcomes

Presenter
Presentation Notes

Combating Multicollinearity
Ridge Regression

• Ridge regression provides an alternative estimation 
method that can be used where multicollinearity is 
suspected

• Logic:
• Multicollinearity leads to small characteristic roots

• When characteristic roots are small, the total mean square error of �β is large 
which implies an imprecision in the least squares estimation method

• Ridge regression gives an alternative estimator (k) that has a 
smaller total mean square error value

Combating Multicollinearity
Ridge Regression

• Ridge Regression for alternative estimator
• The value of k can be estimated by looking at a ridge trace 

plot
• Ridge trace plots are plots of parameter estimates vs k 

where k usually lies in the interval [0,1]

• Note: 
• Pick the smallest value of k that produces a stable estimate of β
• Get the variance inflation factors (VIF) close to 1

Combating Multicollinearty
Ridge Regression Example

• Applying Ridge Regression:
• Use PROC REG procedure with RIDGE option
• RIDGEPLOT option will give graph of ridge trace

/* Ridge Regression Example */

proc reg data=health outvif plots(only)=ridge(unpack VIFaxis=log) 

outest=rrhealth ridge=0 to 0.10 by .002;
model cholesterolloss = age weight cholesterol triglycerides hdl ldl height;

plot / ridgeplot nomodel nostat;

title 'Health - Ridge Regression Calculation';

run;

proc print data=rrhealth;

title 'Health - Ridge Regression Results';

run;

Combating Multicollinearity
Ridge Regression Example

• Choose your alternative estimator
• Pick the smallest value of k that process a stable estimate of β
• Get the variance inflation factors (VIF) close to 1

proc reg data=health outvif plots(only)=ridge(unpack VIFaxis=log) 

outest=rrhealth_final ridge=0 to 0.002 by 0.00002;
model cholesterolloss = age weight cholesterol triglycerides hdl ldl height;
plot / ridgeplot nomodel nostat;
title 'Health - Ridge Regression Calculation';

run;

proc print data=rrhealth_final;
title 'Health - Ridge Regression Results';

run;

Combating Multicollinearity
Ridge Regression Example

• Choose your alternative estimator
• Pick the smallest value of k that process a stable estimate of β
• Get the variance inflation factors (VIF) close to 1

proc reg data=health outvif plots(only)=ridge(unpack VIFaxis=log) 
outest=rrhealth_final ridge=0.00012;
model cholesterolloss = age weight cholesterol triglycerides hdl ldl height;
plot / ridgeplot nomodel nostat;
title 'Health - Ridge Regression Calculation';

run;

proc print data=rrhealth_final;
title 'Health - Ridge Regression Results';

run;

Combating Multicollinearity
Ridge Regression Example

• Modify Output for Interpretation
• Standard errors (SEB)
• Parameter Estimates

proc reg data=health outvif plots(only)=ridge(unpack VIFaxis=log) 
outest=rrhealth_final outseb ridge=0.00012;
model cholesterolloss = age weight cholesterol triglycerides hdl ldl height;
plot / ridgeplot nomodel nostat;
title 'Health - Ridge Regression Calculation';

run;

proc print data=rrhealth_final;
title 'Health - Ridge Regression Results';

run;

Combating Multicollinearity
Ridge Regression Example

After outseb

Before outseb

Summary

• When multicollinearity is present in data
• Ordinary least squares estimators are imprecisely estimated
• This could result in misleading or improper conclusions

• If your goal is to understand how your predictors impact 
your outcome
• Then multicollinearity poses a problem
• Therefore, it is essential to detect and solve this issue before 

estimating the parameters based on the fitted regression model

• The detection of multicollinearity is important

Conclusions

• Once multicollinearity is detected
• Necessary to introduce appropriate changes in model 

specification to combat

• Remedial measures can help solve this problem
• Removing a variable
• Principal Component Regression
• Regularization Techniques

• L1: Lasso Regression
• L2: Ridge Regression



SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

#SASGF

Conclusions

• Remember the Trade-Off?
• Ridge Regression is still an adjustment
• Naturally results in biased outcomes

• Elastic Nets / Bootstrapping
• Could help resolve L1/L2 debate
• Could help address adjustment concerns

Thank You!!

Name: Deanna Schreiber-Gregory
Organization: Henry M Jackson Foundation
Title: Data Analyst, Research Associate
Location: Bethesda, MD
E-mail: d.n.schreibergregory@gmail.com
On LinkedIn

